
 

$regfile = "ATtiny861.DAT" 
$crystal = 1000000 
 
Config Porta = &B11000000                                   'PortA.0-5=input'  PortA.67=output 
Porta = &H58 
'Porta.3 = 1                                                 'プルアップ 
'Porta.4 = 1                                                 'プルアップ 
'Porta.5 = 0                                                 'プルアップ 
'Porta.6 = 1                                                 'ブザー停止 
'Porta.7 = 0                                                 '放電停止  
 
Config Portb = Output 
Portb = &H80 
 
Dim Charge_sw As Byte 
Dim Discharge_sw As Byte 
 
Dim Vsi As Integer                                          '太陽電池電圧 l  
Dim Vs As Long                                              '太陽電池電圧 
Dim Vs_shou As Integer 
Dim Vs_amari As Integer 
 
Dim Vbi As Integer                                          '電池電圧 l  
Dim Vb As Long                                              '鉛電池電圧 
Dim Vb_shou As Integer 
Dim Vb_amari As Integer 
 
Dim Vb_pre_ch As Byte 
Dim Vb_pre_disch As Byte 
 
Dim Vii As Integer                                          '電流値 
Dim I As Long                                               '電流 *10mA 
Dim Iave_sum As Long                                        '平均用合計 
Dim Iave(10) As Word                                        '平均用ﾃﾞｰﾀｰ 
Dim Iave_p As Integer                                       '平均ﾃﾞｰﾀｰ用ポインタ 
Dim Iaverage As Integer                                     '１０回平均電流 
Dim Iave_10 As Byte                                         '10 回読み込み済み 
 
Dim I_shou As Integer 
Dim I_amari As Integer 
 
Dim Mas As Long                                             '放電電流積算 秒 
Dim Mah As Long                                             '放電電流積算 時間ｍAh 
 
Dim T As Long                                               '充放電秒  
Dim Tm As Long                                              '分 
 
Dim Charge_pre As Byte                                      '充電 SW 前の状態 
Dim Chargechu As Byte                                       '充電中信号  
Dim T_teishi As Integer                                     '充電休止 Tm  
Dim T_teishi_up As Byte                                     '充電休止完了 
Dim Disch_pre As Byte                                       '放電 SW 前の状態 
Dim Dischargechu As Byte                                    '放電中信号 
Dim Charge_end As Byte                                      '充電完了信号 
 
Dim On_duty As Integer                                      'PWM オン幅  
Dim Disp_c As Integer                                       'PWM,mAh,Tm 表示切り換え 
 
Declare Sub Battery                                         ' 電池電圧、電流測定サブルーチン 
Declare Sub Lcd_disp                                        '液晶表示サブルーチン 
Declare Sub Buzzer 
 



'========== PWM 設定 ========== 
Tccr1a = &B1000_0010                                        'fast-PWM OC1A 有効 
Tccr1b = &B0000_0001                                        '1/64 に分周 
Tccr1c = &B1000_0000                                        'shadow bits and fast-PWM OC1D 
Tccr1d = &B0000_0000                                        'WGN11 WGM10 =0:高速 PWM 
 
Tc1h = &B0000_0000 : Tcnt1 = &B1111_1111                    'top in 8-bit mode 
 
   Tccr1a = &B0000_0000                                     'PWM 停止 
   Portb.1 = 0                                              'PWM 出力オフ 
 
'==========液晶表示設定========== 
Config Lcdmode = Port                                       ''LCD をポートモードに設定 
Config Lcdbus = 4                                           '制御はポートの上位 4 本を使用  
'---------- LCD 制御ビット指定 --------- 
Config Lcdpin = Pin , Db4 = Portb.3 , Db5 = Portb.2 , Db6 = Portb.6 , Db7 = Portb.0 
Config Lcdpin = Pin , E = Portb.4 , Rs = Portb.5 
Config Lcd = 16 * 2                                         'LCD ﾀｲﾌﾟ指定（16 文字 2 行） 
Display On                                                  '液晶表示開始 
Cursor Off                                                  'カーソル消去 
 
 
 
Config Adc = Single , Prescaler = Auto , Reference = Internal_2.56_nocap       'AD 変換設定_基準電圧
2.56V 外部コンデンサ（PortA.0) 
 
'========== 1 秒タイマ割り込み設定 ========== 
Config Timer0 = Timer , Prescale = 64 , Clear Timer = 1 , Compare A = Disconnect 
   Tccr0a = &B1000_0000                                     '16 ビット TM に設定 
 
'   Stop Timer0 
 
   $asm 
      push R16 
      Push R17 
      LDI  R17,$c1                                          '上位ﾊﾞｲﾄ値 
      LDI  R16,$BF                                          '下位ﾊﾞｲﾄ値 
      Out Tcnt0h , R17 
      Out Tcnt0l , R16 
      pop R17 
      pop R16 
   $end Asm 
 
On Compare0a Timer0_int                                     '割り込み飛び先指定 
Enable Interrupts                                           '割り込み有効 
Enable Compare0a                                            '比較器 1 割り込み有効  
'Start Timer0 
 
'************************* 
'********** Main ********** 
'************************* 
 
St_batterycontrol: 
 
'******************************** 
'********** 充電開始チェック ******* 
 
'---------- SW 状態読み込み ---------- 
   Charge_sw = Pina.3 
   Discharge_sw = Pina.4 
 
   If Chargechu = 1 And T_teishi_up = 0 Then Goto Charge    '充電中なら充電へジャンプ 
'   If Chargechu = 1 Then Goto Charge    '充電中なら充電へジャンプ 



'---------- オンエッジ検出 -------- 
    If Charge_sw = 0 And Charge_pre = 1 Or T_teishi_up = 1 And Vs_shou > 14 Then 
'    If Charge_sw = 0 And Vs_shou > 14 And Charge_pre = 1 Or T_teishi_up = 1 And Vs_shou > 14 Then 
      Charge_pre = Charge_sw 
      Disp_c = 1                                            '表示切り換えカウンター 
      Call Buzzer 
      Chargechu = 1                                         '充電中信号オン  
      Charge_end = 0                                        '充電完了信号クリヤ 
      T_teishi = 0                                          '休止 Tm 
   If Dischargechu = 1 Or T_teishi_up = 1 Then Goto Skip1 
      Mas = 0                                               '放電量クリヤ 
      T = 0                                                 '充放電時間クリヤ  
      Iave_p = 1                                            '平均データーポインタ 初期化 
      Iave_10 = 0                                           '平均済み信号 
Skip1: 
      T_teishi_up = 0                                       '充電休止完了クリヤ 
      Vb_pre_ch = 0                                         '完了電圧回数 
 
  '---------- PWM スタート(  Portb.1) ----------・ 
      On_duty = 10                                          '35                                              
'PWM 初期値 約 20%オン 
      Ocr1a = On_duty 
      Tccr1a = &B1000_0010                                  'PWN 始動 
 
      Goto Charge 
 
   Else 
      Charge_pre = Charge_sw 
   End If 
 
   Goto Discharge_top                                       '放電チェックへ 
 
'****************************** 
'********** 充電制御   **********   
Charge: 
 
'========== 充電中止？ ========== 
 '---------- オンエッジ検出 -------- 
   If Charge_sw = 0 Or Vs_shou < 14 And Charge_pre = 1 Then '放電中止チェック 
      Charge_pre = Charge_sw 
      Tccr1a = &B0000_0000                                  'PWM 停止 
      Portb.1 = 0                                           'PWM 出力オフ 
'    Portb.1 = 0                                            ' 充電停止   
      Chargechu = 0                                         '充電中オフ  
     Call Buzzer 
      Goto Discharge_top 
  Else 
      Charge_pre = Charge_sw 
 
  End If 
'========== 充電完了？========== 
'充電完了チェックは非充電時なので 1 秒割り込みの中で行う。 
'   If Charge_end = 1 Then Call Buzzer 
 
 
 
Discharge_top: 
'******************************** 
'********** 放電開始チェック ******* 
 
   If Dischargechu = 1 Then Goto Discharge                  '放電中なら放電へジャンプ 
 
'---------- オンエッジ検出 ---------- 



   If Discharge_sw = 0 And Disch_pre = 1 Then 
      Disch_pre = Discharge_sw 
      Disp_c = 1 
      Call Buzzer 
      Dischargechu = 1 
      Porta.7 = 1                                           ' 放電開始  
   If Chargechu = 1 Then Goto Skip2 
      Mas = 0                                               '放電量クリヤ 
      T = 0                                                 '充放電時間クリヤ 
      Iave_p = 1                                            '平均データーポインタ 初期化 
      Iave_10 = 0                                           '平均済み信号 
Skip2: 
      Vb_pre_disch = 0 
      Goto Discharge 
   Else 
      Disch_pre = Discharge_sw 
   End If 
 
   Goto St_batterycontrol                                   '最初へ戻る 
 
'****************************** 
'********** 放電開始   **********   
 
Discharge: 
 
'========== 放電中止？========== 
 
'---------- オンエッジ検出 ---------- 
   If Discharge_sw = 0 And Disch_pre = 1 Then               '放電中止チェック 
         Disch_pre = Discharge_sw 
         Porta.7 = 0                                        ' 放電停止 
         Dischargechu = 0                                   '放電終了 
         Call Buzzer 
      Else 
         Disch_pre = Discharge_sw 
   End If 
 
'========== 放電終了？========== 
   If Vb > 1200 Then 
      Vb_pre_disch = 0 
   Else 
'---------- 終了２度チェック 
      If Vb_pre_disch = 0 Then 
         Vb_pre_disch = 1 
      Else                                                  ' 放電終了  
         Porta.7 = 0                                        ' 放電停止 
         Dischargechu = 0 
         Call Buzzer 
      End If 
   End If 
 
   Goto St_batterycontrol                                   '最初へ戻る 
 
  End 
 
 
'*************************************** 
'********** サブルーチン・***************** 
'*************************************** 
 
 
'*************************************** 
'********** 1 秒割り込みプログラム ********** 



Timer0_int: 
 
'----------タイマ初期化---------- 
  Stop Timer0 
  Disable Compare0a 
   $asm 
      push R16 
      Push R17 
      LDI  R17,$c1                                          '上位ﾊﾞｲﾄ値  設定値 499111 の 16 進数 
&HC2F7 の上位バイト 
      LDI  R16,$BF                                          '下位ﾊﾞｲﾄ値 
      Out Tcnt0h , R17                                      ' 経過値上位書き込み  
      Out Tcnt0l , R16                                      ' 経過値下位書き込み  
      pop R17 
      pop R16 
   $end Asm 
  Enable Compare0a                                          '比較器 1 割り込み有効  
  Start Timer0                                              '1 秒割り込みタイマースタート 
 
'---------- 充放電電圧電流測定-------- 
   Call Battery                                             ' 電池測定サブルーチン  
 
   If Chargechu = 1 And Charge_end = 0 Or Dischargechu = 1 Then 
         If Pina.5 = 1 Then Mas = Mas - I Else Mas = Mas + I       '放電量積算 PINB.6:極性信 
'         Mas = Mas + I                                      '放電量積算 
         T = T + 1                                          '秒積算 
         Tm = T ¥ 60                                        '分 
         Mah = Mas / 360                                    'ｍAh 換算 （ｍAs*10/3600・ 
      Else 
 
   End If 
 
'---------- 休止中 Tm ---------- 
   If Chargechu = 1 And Charge_end = 1 Then T_teishi = T_teishi + 1       '休止 Tm 加算 
   If T_teishi > 1800 Then 
      T_teishi_up = 1                                       '充電休止完了  
      Charge_end = 0                                        '充電完了クリヤ 
   End If 
 
'========== PWM 調整 ========== 
   If Chargechu = 0 Then Goto Tint_end 
'   If Chargechu = 0 Or Charge_end = 1 Then Goto Tint_end    '非充電時又は休止中ならジャンプ 
'----------MPPT 制御 ---------- 
'太陽電池の最大電力取り出しの為、最大電力取り出し電圧 17.2V になる様電流量を PWM で調整。 
   If Vs > 1750 Then On_duty = On_duty + 1                  '太陽電池電圧が高ければ、電流を増やす。 
         If On_duty > 255 Then On_duty = 255 
   If Vs < 1700 Then On_duty = On_duty - 10                 '太陽電池電圧が低ければ、電流を減らす 
         If On_duty < 1 Then On_duty = 1 
      Ocr1a = On_duty                                       'PWM 設定値書き込み  
 
'---------- 充電完了？ ---------- 
'   If Vbstop < 1360 Then                                    '充電完了？ 
   If Vb < 1480 Then                                        '充電完了？ 
      Vb_pre_ch = 0                                         '完了電圧回数 
      Goto Tint_end                                         '最後へ 
   Else                                                     '充電終了電圧より高い時 
   '---------- 終了 2 度チェック --------- 
         If Vb_pre_ch = 0 Then 
            Vb_pre_ch = 1 
            Goto Tint_end 
         Else                                               '2 度目なら 
            Charge_end = 1                                  '充電完了  
            Tccr1a = &B0000_0000                            'PWM 停止 



            Portb.1 = 0                                     'PWM 出力オフ 
'            Chargechu = 0                                   '充電中オフ  
            Call Buzzer 
         End If 
 
   End If 
 
 
Tint_end: 
 
 '---------- 電圧電流表示-------- 
   Call Lcd_disp                                            '液晶表示 
 
 
 Return 
 
'********************************************** 
'********** 電池電圧電流測定サブルーチン  *********  
 Sub Battery 
 
   Start Adc                                                'A/D コンバータに電源を供給。 
'---------- 鉛電池電圧読み込み 
      Vbi = Getadc(0)                                       '電圧値読み込み。 
      Vb = Vbi * 2560                                       '電池電圧換算（分圧値を 10 倍）  
      Vb = Vb / 1023 
      If Vb < 2 Then Vb = 0                                 '０V 誤差削除 
      Vb_shou = Vb ¥ 100 
      Vb_amari = Vb Mod 100 
 
'---------- 電流読み込み 
      Vii = Getadc(1)                                       '電流値読み込み 
      I = Vii * 25600                                       'AD 入力電圧、電池には＊４ 
      I = I / 51150 
'      If I < 2 Then I = 0                                   '０A 誤差削除 
      I = I - 13                                            '10 増幅時０V 誤差補正 
'１０回の平均電流算出値 
      Iave_sum = Iave_sum - Iave(iave_p)                    '一番古いﾃﾞｰﾀｰ削除 
      Iave(iave_p) = I                                      '新しいデータ追加 
      Iave_sum = Iave_sum + I                               '新しいﾃﾞｰﾀｰ加算 
 
      '平均データーポインタ＋１  
      Iave_p = Iave_p + 1 
      If Iave_p > 10 Then 
         Iave_10 = 1                                        '初回平均完了 
         Iave_p = 1                                         '平均データーポインタ初期化 
      End If 
 
      Iaverage = Iave_sum / 10                              '電流平均値 
 
      '正しい平均値が出るまでは瞬時値電流表示 
      If Iave_10 = 0 Then I = I Else I = Iaverage 
'整数部と小数点部に分ける 
      I_shou = I ¥ 100                                      '整数部  
      I_amari = I Mod 100                                   '少数部  
 
'---------- 太陽電池電圧読み込み 
      Vsi = Getadc(2) 
      Vs = Vsi * 2560                                       '太陽電池電圧換算（分圧値を 10 倍）  
      Vs = Vs / 1023 
      Vs_shou = Vs ¥ 100 
      Vs_amari = Vs Mod 100 
 
   Stop Adc 



 
 
 End Sub 
 
 
'******************************* 
'**********液晶表示制御********** 
 
 Sub Lcd_disp 
 
 Cls                                                        '表示クリヤ  
 
'----------上の行表示設定  
'---------- 太陽電池電圧表示  
   Locate 1 , 1 
      Lcd "S " 
   If Vs_shou < 10 Then Locate 1 , 3 Else Locate 1 , 2 
       Lcd Vs_shou 
   If Vs_amari < 10 Then 
         Locate 1 , 4 : Lcd ".0" : Lcd Vs_amari 
      Else 
         Locate 1 , 4 : Lcd "." : Lcd Vs_amari 
      End If 
   Locate 1 , 7 
      Lcd "V" 
'---------- 電流表示 
   Locate 1 , 9 
      If Pina.5 = 1 Then Lcd "I=-" Else Lcd "I= " 
         Lcd I_shou 
      If I_amari < 10 Then 
         Locate 1 , 13 : Lcd ".0" : Lcd I_amari 
      Else 
         Locate 1 , 13 : Lcd "." : Lcd I_amari 
      End If 
   Locate 1 , 16 
      Lcd "A" 
 
'----------下の行表示設定・ 
   Disp_c = Disp_c + 1 
      If Disp_c > 2 Then Disp_c = 1 
 
   Locate 2 , 1 
 
   Select Case Disp_c 
      Case 1 : 
'---------- 鉛電池電圧  
         Lcd "B " 
            If Vb_shou < 10 Then Locate 2 , 3 Else Locate 2 , 2 
               Lcd Vb_shou 
            If Vb_amari < 10 Then 
               Locate 2 , 4 : Lcd ".0" : Lcd Vb_amari 
            Else 
               Locate 2 , 4 : Lcd "." : Lcd Vb_amari 
            End If 
         Lcd "V" 
'---------- 充放電量表示  
         Locate 2 , 9 : Lcd Mah 
         Locate 2 , 14 : Lcd "mAh" 
 
      Case 2 : 
'---------- PWM 値表示  
         Lcd "PWM=" : Lcd On_duty 
 



'---------- 充放電時間表示  
         Locate 2 , 9 
            Lcd "T=" : Lcd Tm : Locate 2 , 16 : Lcd "m" 
   End Select 
 
 
End Sub 
 
 
'********** ブザー鳴らし ********** 
Sub Buzzer 
        Porta.6 = 0                                         'ブザー鳴動 
           Waitms 100 
        Porta.6 = 1                                         'ブザー停止 
End Sub 


